AD-A118 227 DEFENCE RESEARCH ESTABLISHMENT OTTAWA (ONTARIO) F/G 9/2
A COMPUTER PROGRAM IN FINNIGAN BASIC FOR PLOTTING REPORT-QUALIT--ETC(U)
MAR 81 J G PURDON, J M PRESTON
UNCLASSIFIED DREO-TN-82-3 NL





RESEARCH AND DEVELOPMENT BRANCH

DEPARTMENT OF NATIONAL DEFENCE
CANADA

DEFENCE RESEARCH ESTABLISHMENT OTTAWA

TECHNICAL NOTE NO. 82-3

A COMPUTER PROGRAM IN FINNIGAN BASIC FOR PLOTTING REPORT-QUALITY GRAPHS USING A FINNIGAN 6115 DATA SYSTEM

by

J.G. Purdon and J.M. Preston
Chemical Defence Section
Protective Sciences Division

MARCH 1981
OTTAWA

ABSTRACT

A BASIC computer program has been developed for drawing either half- or full-page-size report-quality graphs of externally generated data on a Finnigan 4010 Gas Chromatograph/Mass Spectrometer Data System. Up to five sets of results can be linearly, logarithmically or inversely presented. As well as drawing the borders and printing labels supplied by the user, the program selects appropriate divisions for the axes and scales the data accordingly. Either solid straight lines joining all points or solid or dashed straight lines resulting from an internal linear least squares calculation can be included on the graph. Provision has also been made for the inclusion of threshold lines on the graph if desired and for the separate printing of lines of alphanumeric and plotting characters for use in captions or headings. A listing, flow-chart and set of simplified instructions for its use are included.

RÉSUMÉ

Un programme informatique BASIC a été mis au point pour effectuer le traçage de graphiques en format demi-page ou pleine page, obtenus à partir de données compilées sur le système informatique d'un ensemble Chromatographe gazeux/Spectromètre de masse Finnigan 4010. Au maximum cinq séries de résultats peuvent être présentées, sous forme linéaire, logarithmique ou inverse. Outre le traçage des marges et l'impression des labels fournis par l'utilisateur, le programme choisit les divisions appropriées pour les axes et cadre les données en conséquence. Le graphique peut comporter des lignes droites continues reliant tous les points ou des lignes droites continues ou en pointillées obtenues d'après un calcul linéaire interne des moindres carrés. On a également prévu l'inclusion de lignes limites, sur le graphique, si désirées, et l'impression séparée de lignes en caractères alphanumériques ou en caractères de traçage, pour utilisation dans les légendes et les titres. Son inclus une liste imprimée, un organigramme et des instructions vulgarisées.

INTRODUCTION

In many technical reports, numerical results are generated which can be represented as functions of one or more independent variables or conditions. Often, the response of a measured quantity is best presented in graphical form, especially if the theoretical or expected response has been postulated and can similarily be presented. On some occasions, an author may wish to view more than one presentation of the same material before deciding which one best emphasizes a particular dependence of interest. The advantages of graphical presentation result in increased demand on drafting and graphics facilities.

Many dedicated computer systems supplied with modern laboratory instrumentation possess the capability of plotting graphs of results generated on the specific instrument. More sophisticated systems also permit the user to write separate programs using the resident software to drive the peripheral equipment. In the case of the Finnigan 4010 Gas Chromatograph/Mass Spectrometer (GC/MS) equipped with a 6115 Data System (DS), the user may develop BASIC software to drive the plotter provided with the system.[1] We have developed a program (PLOTTR) which permits inexperienced personnel to plot report-grade graphs of their results after a brief instruction on the operation of the system. A listing and flowchart of the program are given in Appendices I-II and a simplified set of instructions is given in Appendix III.

PROGRAM PLOTTR

GENERAL

PLOTTR will draw a graph with numbers and axes labelled on full- or half-size blue-line paper. Each axis can be plotted as a linear, logarithmic (base 10) or inverse function. The appearance and readability are improved if the numbers to be printed on an axis are adjusted to three digits, in the case of linear or inverse axes, by the selection of suitable units. Each axis is marked by tick marks at each printed number or by a system of short and long tick marks depending on the number of numerical labels to be included on the axis. Alphanumeric labels, of up to a maximum number of characters determined by the axis length in inches are accepted, centered, and printed along the appropriate axis in the normal manner.

The program will plot up to five different sets of results of up to 100 points each (the number of points could be increased subject to memory limits by simple modification of the program) using the symbols available. Each data set is entered, in turn, in the form (x, y) with provisions made for the re-entering of incorrectly entered numbers and for detection of the end of a given data set. Following entry of all data sets, the minimum and maximum values of x and y are found and the axis scaled so that the presentation fills the graph. If the user wishes to extend the graph beyond the calculated range, e.g. to have the graph include the origin even though no data points are located there, this can be achieved by entry of a separate dummy data set (respecting the upper limit of five sets of data) consisting

of two data points located at the limits of the graph desired. By suppressing the plotting of the two points, the full graph will be drawn but only the original data sets plotted.

The user has the option of choosing a solid line to join each point in a data set; of having a linear least-squares calculation performed, constants output by typewriter and plotted as a solid or long- or short-dashed line or of having no line at all if the data points describe a curve which is to be manually added after the plotting operation. As the plotting of one or more of the data sets can be suppressed if desired, the user who wishes to draw, for example, a threshold indicated by a straight horizontal or sloped line can easily do so by entering two points on the threshold line as a distinct data set, suppressing plotting of the individual points, calling for a linear least squares calculation, and plotting the result as a solid or dashed line.

Provision has also been made to produce more than one copy of a given graph without re-entry of the data. A final option allows the printing, away from the graph, of any number of lines of alphanumerics in the same format as the axis labels for such use as captions or equations of lines. As part of this option, a line of the symbols which have been used for plotting the individual points can also be printed for inclusion in a caption.

Whenever input from the user is required, a message is printed at the typewriter stating just what is required. These prompting messages were written with the aim that no other manual or instructions would be required, i.e. the program would 'talk the user through' plotting the graph.

THE PROGRAM

The program consists of three parts: the main body which accepts the input, calculates the least squares[2], draws the graph borders, and plots the points and lines; and two subroutines, one for linear and inverse presentations, and one for logarithmic plots. These subroutines calculate the scaling of the entire graph and the selection and positioning of numerical labels and tick marks on each axis. A number of resident sub-routines, specific to the Finnigan BASIC system are called upon to operate the plotter for plotting lines and points and printing letters and numbers. A more detailed description of the program logic follows:

THE MAIN BODY

Statements 70-560, 2183-2191, 4950-5080 - These sections accept the user's selections of the various output options available and the sets of data to be presented on the graph. The operations of these sections are covered in Appendix III in the simplified instructions for use of PLOTTR.

Statements 572-6 - This section defines the functions LOG(x), 1/x and two others which are often used by subprograms 1 and 2 (SP1 and SP2).

Statements 584-610 - These statements redefine the co-ordinates of the points in terms of the selected function, i.e. no change, or replacing the variable by its logarithm (base 10) or inverse.

Statements 611-810 - In this section, data points for each set are rearranged into ascending order of x to simplify plotting operations and the maxima and minima of all the x and y values are found.

Statements 1010-1260 - In this section, requested linear least-squares calculations[2] are carried out on data sets and the data set number, the number of points, average x, average y, slope and intercept and their standard deviations and the correlation coefficient are printed on the system typewriter (the last three are suppressed if there are only two points in the data set).

Statement 1265 - The pen is located at the origin of the graph.

Statements 1280-1460 - The borders and axis labels are plotted according to the choice of full-size or half-size blue-line paper.

Statements 1470-1728 - The appropriate subprogram (SP1 for linear/inverse, SP2 for logarithmic) is called, once for each axis, to calculate the axis subdivision and plot the tick marks and numerical labels.

Statements 1730-1820 - The individual data points are added to the graph after the coordinates have been scaled to the ranges used in the axis subdivision. The variable H9 is used to bypass the scaling operation if the graph is a second or subsequent copy.

Statements 1830-2179 - This section draws the requested lines on the graph. The option of joining all points by straight lines is accomplished by statements 1860-1910 while the remainder plots the results of a least squares calculation. Values are calculated for X7, Y7 and X8, Y8, the ends of the line for the graph, in section 1940-2140 to ensure that the plotted line possesses the correct slope and stops at the axis no matter where the intersection is. The resultant values of X7, Y7, X8 and Y8 are scaled to the ranges of the axes in statements 2150-3. Again, if the graph being drawn is a second or subsequent copy, part of this calculation is bypassed (H9 > 0, statement 1742). If a solid line is to be dashed section 2159-74 is utilized. In order that a dashed line be drawn with the correct slope from border to border yet retain a constant dash length, the simultaneous solution to equations 1 and 2 is used in which the slope (a) and intercept (b) are scaled to the graphical coordinates (2159) and a length of ten graph units is selected as the basic dash length.

$$y = ax + b \qquad (1)$$

$$x^2 + y^2 = 100 \qquad (2)$$

R(I) can be reassigned a value other than 1 or 3 to produce dashes of different lengths. The variable Z7 is used to alternatively raise or lower the pen to achieve the dashed effect.

Statements 2180-91 - Execution of 2180-2 advances the paper to a new page (double advance is required for a large format graph as it extends over two pages) while statements 2183-91 present the option of duplicating a graph.

Statements 4950-5060 - Lines of alphanumeric print for use in captions or as equations for lines on graph as well as a line of plotting symbols can be printed out.

Statements 5064-end - The paper is moved to a new page and program ends.

SUBPROGRAM 1 (SP1, statements 2990-3380)

This subroutine designs and plots the tick marks and numeric labels for one linear axis. In statements 3000-3090, the maximum and minimum values are used to select the increment between major tick marks, D6, and thus find the vectors P and Q, the values and locations of these marks. The next section* (3095-3130) deduces the number of digits in the largest numeric label, A6, allows for a negative sign (M4<0), and then suppresses the decimal point and first decimal place if not required, B6, to find the half-width of the label, W3, in hundredths of an inch. If the labels are wider than the distance between tick marks, only every T6th label is written (statements 3140-3144 ensure that zero is not skipped for this reason). There follow separate instructions for the X-axis (3150-3260) and the Y-axis (3266-3370) for drawing major and minor tick marks on both sides of the graph and writing the numeric labels outside the lower or left-hand sides. The minor tick marks, midway between the major tick marks, are half the height of the latter unless the difference between the major tick marks is ..., 0.2, 2, 20 ,... in which case all marks are equally tall. The values of the ends of the axis, X0 and X9, Y0 and Y9 are computed for later use. Statements 3198-9, 3289-91 and all statements involving E6 were empirically constructed to operate with the Finnigan NUMBER statement, which does not operate sensibly with non-positive numbers.

SUBPROGRAM 2 (SP2, statements 3490-4110)

This subroutine designs and plots the tick marks and numeric labels of one logarithmic axis. The minimum and maximum values, M4 and M5, which have already been converted to their logarithms, are used to compute the total number of tick marks, N6, and the number of tick marks between the edges of the graph and the first and last decades, G6 and H6 (3530-3580). In statements 3500-3520, 3590-3610 and 3720-3770, the vector of numeric labels, P(I), is constructed such that P(I) is the value of the power of ten at decade marks (e.g. 2 for 10^2), one million at intervening marks which will not be labelled, and one hundred plus the value at intervening tick marks which will be labelled, e.g. 107 for 7 or 700. The intervening tick marks are labelled if only one power of ten will be labelled on the axis. The locations of the marks, Q(I), are computed, depending on the length in inches, L6, of the axis (3620-3700). There follow separate instructions for X-axis (3790-3945) and Y-axis (3950-4100) for plotting and labelling the tick marks. Again the values of the ends of the axis are computed, and again it is necessary to allow for odd behaviour of the NUMBER statement with non-positive numbers (the expressions (P(I)<0) in 3850 and 4010).

* In Finnigan BASIC, a bracket expression containing <, > or =, such as (M5 =0), is assigned a value of unity if the bracketed expression is true and zero if it is false.

SYSTEM SUBPROGRAMS

The following subroutines are part of the Finnigan BASIC software package and are included here to indicate the role of each arguments of the subroutine.

1. PLOT A, B, P - This subroutine causes the pen to move to the point (A, B) (coordinates in units of 0.01 inch) while P defines whether the pen is raised or lowered.

2. SYMBOL A, B, H, B$, R, C - This subroutine prints the string B$ of C alphanumeric characters a height of H x 0.07 inches in a direction defined by R (0-3 corresponding to 0^{o}, 90^{o}, 180^{o} and 270^{o}) starting at coordinates (A, B).

3. LINE A, B, P, M, I - This subroutine draws a line through P points defined by the pairs (x,y) in arrays A, B. The other arguments are not used here.

4. NUMBER A, B, H, V, R, N - This subroutine prints the number V a height of H x 0.07 inches in a direction defined by R starting from coordinates A, B. The placing of the decimal is governed by N.

5. MARKER P - This subroutine plots a character to represent a point (x,y) on the graph at the current position of the pen. Five different symbols are available depending on the value of P.

6. LEN (B$) - This subroutine determines the number of members in a string of alphanumeric characters B$.

7. RSTR - The pen is raised and paper moved to a new page. The pen is positioned in a pre-determined location on the new page.

EXAMPLES

Examples demonstrating the use of various program options are given in Figures 1 and 2. In Figure 1, five sets of data, whose numerical values are tabulated in Table 1, are plotted. Set #1 consists of a series of points approximating a straight line. The dashed line is the result of a linear least squares treatment. Set #2 is used to demonstrate the option in which the individual points are joined by solid straight lines while set #3 is given as a set of plotted points on a curve which could later be added manually. Set #4, consisting of only two points, demonstrates the use of PLOTTR to include straight threshold lines by carrying out a linear least squares treatment but suppressing the plotting of the points. In set #5, two points whose output is suppressed are used to extend the upper border of the graph a height of 0.15 units beyond that of any of the plotted points. Figure 2 demonstrates the ability of the program to plot logarithmic functions. The characters for the captions and equations for the straight lines for Figures 1 and 2 were printed on a separate page by the plotter after execution of the plots.

TABLE 1

Test Data for Figure 1

Set	X	Y	Set	X	Y
1	5.0	0.66	3	75.0	0.508
	10.5	0.61		70.0	0.5
	17.0	0.55		65.0	0.49
	22.2	0.5		60.0	0.48
	28.0	0.45		55.0	0.47
	30.0	0.43		50.0	0.46
	35.0	0.39		45.0	0.448
	40.0	0.34		40.0	0.434
	44.0	0.30		35.0	0.418
	49.0	0.25		30.0	0.40
	55.0	0.20		25.0	0.38
	60.0	0.155		20.0	0.35
	65.0	0.11		15.0	0.32
	70.0	0.065		10.0	0.27
2	5.0	0.85		5.0	0.19
	10.0	0.80		2.0	0.08
	12.5	0.825		3.0	0.13
	15.0	0.75		4.0	0.16
	20.0	0.80	4	5	0.525
	30.0	0.70		70	0.525
	40.0	0.85	5	0	0
	45.0	0.70		80	0.99
	55.0	0.85			
	60.0	0.85			
	65.0	0.75			
	70.0	0.85			



FIGURE 1. LINEAR AXIS GRAPH DRAWN BY PLOTTR
SET#1 = X ; SET#2 = ▽ ; SET#3 = +



FIGURE 2. LOGARITHMIC GRAPH DRAWN BY PLOTTR

1. Finnigan BASIC Reference Manual, Revision I, August 1977.

2. Kemeny, J.G. and Kurtz, T.E. BASIC programming. 2nd ed. New York: Wiley, 1971. pp 114-5.

APPENDIX I

PROGRAM LISTING - PLOTTR

```
70   DIM A(6),B(6),C$(60),D$(35),C(101,5,2),E(6),F(6),G(6),P(101),Q(101)
75   DIM H(2),R(6)
80   PRINT"ENTER THE LABEL TO BE USED ALONG THE X-AXIS"
100  INPUT C$
102  PRINT"ENTER THE LABEL TO BE USED ALONG THE Y-AXIS"
108  INPUT D$
130  PRINT"THE NUMBER OF SETS OF DATA ARE (1-5)"
140  INPUT A1
180  PRINT"A SELECTION OF SYMBOLS FOR EACH DATA SET IS MADE FROM THESE:"
185  PRINT
190  PRINT"0=NONE (LINE IS STILL POSSIBLE), 1=CROSSES, 2=X'S"
195  PRINT
200  PRINT"ALSO AN INDICATION IS NEEDED WHETHER A LINE IS DESIRED FOR"
210  PRINT"THAT SET BASED ON: 0=NO LINE, 1=A LINE JOINING ALL POINTS OR 2="
220  PRINT"LEAST SQUARES LINE BASED ON EQUATION Y=AX+B. ALSO, THE LINE"
225  PRINT"FOR LEAST SQUARES CAN BE 0=SOLID,1=DASHES OR 3=LONG DASHES"
230  PRINT"ANSWER BY ENTERING A CHOICE FOR OPTION 1, FOLLOWED BY A COMMA"
240  PRINT"THE SECOND OPTION,COMMA,AND THE THIRD, THEN 'RETURN'"
245  PRINT
250  FOR I=1 TO A1
260  PRINT"FOR DATA SET #";I;"SELECTED OPTIONS ARE";
270  INPUT A(I),B(I),R(I)
280  NEXT I
285  PRINT
290  PRINT"THE SIZES ARE 1=3.5 IN X IN OR 2=5 IN X 8 IN, ENTER CHOICE";
300  INPUT A2
305  PRINT
310  PRINT"THE CHOICES AVAILABLE FOR AXES;1=LINEAR,2=LOGARITHM(BASE 10)"
312  PRINT"3=INVERSE;INPUT CHOICE FOR X,COMMA,Y,RETURN";
314  INPUT H(0),H(1)
380  LET C1=LEN(C$)
430  LET C2=LEN(D$)
440  PRINT"NEXT THE VALUES OF THE POINTS (X,Y) ARE INPUT IN FORM X,Y"
450  PRINT"FOLLOWED BY RETURN-UP TO 100 DATA POINTS MAY BE ENTERED-IF AN"
460  PRINT"ERROR IS MADE,FINISH ENTRY,ENTER THE NEXT X,Y AS 999999,0"
470  PRINT"THEN REENTER CORRECT PAIR OF NUMBERS-WHEN DATA SET IS"
480  PRINT"FINISHED,ENTER 0,999999 AND PRESS RETURN"
485  PRINT
490  FOR I=1 TO A1
492  PRINT"ENTER THE DATA FOR SET #";I
494  LET G(I)=0
496  FOR J=1 TO 100
500  INPUT C(J,I,0),C(J,I,1)
510  IF C(J,I,0) <> 999999 THEN 530
515  LET G(I)=G(I)-1
520  LET J=J-1
525  GO TO 500
530  IF C(J,I,1)=999999 THEN 560
540  LET G(I)=G(I)+1
550  NEXT J
560  NEXT I
572  DEF FNL (X)=0.4342945*LOG(X)
574  DEF FNI (X)=1/X
575  DEF FNA (X)=INT(FNL(ABS(X)))
576  DEF FNW (X,Y)=7*(1+2*(Y<1)+(X<0)+(X>10)*FNA(X)+(Y<.1)*FNA(.9/Y))
584  FOR N=0 TO 1
586  ON H(N) GO TO 610,588,600
588  FOR K=1 TO A1
590  FOR I=1 TO G(K)
592  LET C(I,K,N)=FNL(C(I,K,N))
594  NEXT I
596  NEXT K
598  GO TO 610
600  FOR K=1 TO A1
```

APPENDIX I

PROGRAM LISTING - PLOTTR

```
602  FOR I=1 TO G(K)
604  LET C(I,K,N)=FNI(C(I,K,N))
606  NEXT I
608  NEXT K
610  NEXT N
611  REM THE FOLLOWING IS THE INTERCHANGE OF VARIABLES INTO ASCENDING X
612  REM AND FINDS THE MINIMUM AND MAXIMUM X AND Y VALUES
613  LET M1=C(1,1,0)
614  LET S1=C(1,1,0)
616  LET M2=C(1,1,1)
618  LET S2=C(1,1,1)
620  FOR K=1 TO A1
630  FOR 1=1 TO G(K)-1
640  FOR J=1+1 TO G(K)
650  IF C(J,K,0)>=C(I,K,0) THEN 712
660  LET X1=C(I,K,0)
670  LET Y1=C(I,K,1)
680  LET C(I,K,0)=C(J,K,0)
690  LET C(I,K,1)=C(J,K,1)
700  LET C(J,K,0)=X1
710  LET C(J,K,1)=Y1
712  NEXT J
714  NEXT I
716  FOR I=1 TO G(K)
720  IF M1 =C(I,K,0) THEN 740
730  LET M1=C(I,K,0)
740  IF<S1 =C(I,K,0) THEN 760
750  LET S1=C(I,K,0)
760  IF M2 >C(I,K,1) THEN 780
770  LET M2=C(I,K,1)
780  IF S2<=C(I,K,1) THEN 800
790  LET S2=C(I,K,1)
800  NEXT I
810  NEXT K
1010 REM THIS CALCULATES THE LEAST SQUARES LINES FOR THE DATA POINTS
1020 FOR I=1 TO A1
1030 IF B(I) <> 2 THEN 1260
1040 LET X1=X2=Y1=Y2=Z=0
1050 FOR J=1 TO G(I)
1060 LET X1=X1+C(J,I,0)
1070 LET Y1=Y1+C(J,I,1)
1080 LET X2=X2+C(J,I,0)*C(J,I,0)
1090 LET Y2=Y2+C(J,I,1)*C(J,I,1)
1100 LET Z=Z+C(J,I,0)*C(J,I,1)
1110 NEXT J
1120 LET X3=X1/G(I)
1130 LET Y3=Y1/G(I)
1140 LET X4=X2-X1*X3
1150 LET Y4=Y2-Y1*Y3
1160 LET X5=Z-X1*Y3
1170 LET E(I)=X5/X4
1180 LET F(I)=Y3-E(I)*X3
1210 PRINT"FOR SET #";I;"NUMBER OF POINTS =";G(I);"MEAN OF X=";X3;
1212 PRINT"OF Y=";Y3
1220 PRINT"Y= ";E(I);"*X +(";F(I);")"
1222 IF G(I)<=2 THEN 1260
1224 LET Y5=Y4-E(I)*X5
1226 LET X6=Y5/(G(I)-2)
1230 PRINT"STANDARD DEVIATION OF SLOPE = ";SQR(X6/X4)
1235 LET X1=SQR(X6*(1/G(I)+X3*X3/X4))
1240 PRINT"                    OF Y-INTERCEPT = ";X1
1250 PRINT "CORRELATION OF COEFFICIENT=";X5/SQR(X4*Y4)
1260 NEXT I
1265 PLOT 100,100,-1
1270 LET H9=0
```

APPENDIX I

PROGRAM LISTING - PLOTTR

```
1280 REM THIS PRINTS OUT BORDERS AND PRINTS LABELS ON AXES
1310 IF A2<2 THEN 1390
1320 PLOT 800,0,2
1330 PLOT 800,500,2
1340 PLOT 0,500,2
1350 PLOT 0,0,2
1360 LET C3=400-7*C1
1370 LET C4=250-7*C2
1380 GO TO 1450
1390 PLOT 500,0,2
1400 PLOT 500,350,2
1410 PLOT 0,350,2
1420 PLOT 0,0,2
1430 LET C3=250-7*C1
1440 LET C4=175-7*C2
1450 SYMBOL C3,-65,2,C$,0,C1
1460 SYMBOL -65,C4,2,D$,1,C2
1470 REM THIS CALLS ON SUBROUTINE TO SUBDIVIDE AND MARK AXES
1480 IF A2<2 THEN 1620
1490 LET L6=800
1500 LET M6=500
1510 LET M4=S1
1520 LET M5=M1
1530 LET R6=1
1532 ON H(0) GO TO 1536,1540,1536
1536 GO SUB 3000
1538 GO TO 1550
1540 GO SUB 3500
1550 LET L6=500
1560 LET M6=800
1570 LET M4=S2
1580 LET M5=M2
1590 LET R6=2
1592 ON H(1) GO TO 1596,1600,1596
1596 GO SUB 3000
1598 GO TO 1740
1600 GO SUB 3500
1602 GO TO 1740
1620 LET L6=500
1630 LET M6=350
1640 LET M4=S1
1650 LET M5=M1
1660 LET R6=1
1662 ON H(0) GO TO 1664,1668,1664
1664 GO SUB 3000
1666 GO TO 1680
1668 GO SUB 3500
1680 LET L6=350
1690 LET M6=500
1700 LET M4=S2
1710 LET M5=M2
1720 LET R6=2
1722 ON H(1) GO TO 1724,1728,1724
1724 GO SUB 3000
1726 GO TO 1740
1728 GO SUB 3500
1730 REM THIS SECTION PLOTS THE POINTS ON THE GRAPH
1740 DOR J=1 TO A1
1742 IF H9>0 THEN 1750
1743 FOR I=1 TO G(J)
1744 LET C(I,J,0)=(C(I,J,0)-X0)/(X9-X0)*M6
1745 LET C(I,J,1)=(C(I,J,1)-Y0)/(Y9-Y0)*L6
1746 NEXT I
1750 IF A(J)=0 THEN 1820
1760 FOR I=1 TO G(J)
```

APPENDIX I

PROGRAM LISTING - PLOTTR

```
1790 PLOT C(I,J,0),C(I,J,1),3
1800 MARKER A(J)
1810 NEXT I
1820 NEXT J
1830 REM THIS SECTION PUTS THE STRAIGHT LINES ON THE GRAPH
1840 FOR I=1 to A1
1850 ON B(I)+1 GO TO 2179,1860,1930
1855 REM THIS LINE JOINS ALL THE PLOTTED POINTS
1860 FOR J=1 TO G(I)
1870 LET P(J-1)=C(J,I,0)
1880 LET Q(J-1)=C(J,I,1)
1890 NEXT J
1900 LINE P,Q,G,(I),0,0
1910 GO TO 2179
1920 REM THIS LINE IS THE RESULT OF LEAST SQUARES ANALYSIS
1930 LET Y7=E(I)*X0+F(I)
1935 LET Z7=1
1940 IF Y7<Y0 THEN 1960
1950 IF Y7<Y9 THEN 2030
1960 IF E(I)>0 THEN 2000
1970 LET X7=(Y9-F(I))/E(I)
1980 LET Y7=Y9
1990 GO TO 2040
2000 LET X7=(Y0-F(I))/E(I)
2010 LET Y7=Y0
2020 TO TO 2040
2030 LET X7=X0
2040 LET Y8=E(I)*X9+F(I)
2050 IF Y8>Y9 THEN 2070
2060 IF Y8>Y0 THEN 2140
2070 IF E(I)>0 THEN 2110
2080 LET X8=(Y0-F(I))/E(I)
2090 LET Y8=Y0
2100 GO TO 2150
2110 LET X8=(Y9-F(I))/E(I)
2120 LET Y8=Y9
2130 GO TO 2150
2140 LET X8=X9
2150 LET X7=(X7-X0)/(X9-X0)*M6
2151 LET Y7=(Y7-Y0)/(Y9-Y0)*L6
2152 LET X8=(X8-X0)/(X9-X0)*M6
2153 LET Y8=(Y8-Y0)/(Y9-Y0)*L6
2154 IF R(I)>0 THEN 2159
2155 PLOT X7,Y7,3
2156 PLOT X8,Y8,2
2157 GO TO 2179
2159 LET S3=(Y8-Y7)/(X8-X7)
2160 PLOT X7,Y7,3
2161 LET W7=X7
2162 LET W8=Y7
2163 LET L=(1/(1+S3^2))
2166 LET W7=W7+SQR(L)*10*R(I)
2167 LET W8=W8+S3*SQR(L)*10*R(I)
2168 IF W7>X8 THEN 2179
2169 PLOT W7,W8,2+(Z7=2)
2170 ON Z7 GO TO 2171,2173
2171 LET Z7=2
2172 GO TO 2166
2173 LET Z7=1
2174 GO TO 2166
2179 NEXT I
2180 IF A2<2 THEN 2182
2181 RSTR
2182 RSTR
2183 PRINT "IS ANOTHER COPY DESIRED? 0=NO; 1=YES";
```

APPENDIX I

PROGRAM LISTING - PLOTTR

```
2184 INPUT H9
2185 IF H9>0 THEN 1310
2191 GO TO 4950
2990 REM THIS SECTION USES THE EXTREME VALUES OF THIS AXIS TO DETERMINE
2991 REM THE FORM OF DISPLAY AND THE INTERVAL BETWEEN LABELS. P IS THE
2992 REM ARRAY OF VALUES AND Q THE CORRESPONDING POSITIONS, W3 IS THE
2293 REM HALF-WIDTH OF THE LABEL
3000 LET P6=FNA (M5-M4)
3010 LET G6=INT((M5-M4)*10^(-1*P6))
3020 LET G6=G6+(10-G6)*(G6>2)
3025 LET D6=G6*10^(P6-1)
3030 LET N6=INT(M5/D6)-INT(M4/D6)+2
3040 LET P(1)=D6*INT(M4/D6)
3050 LET Q(1)=0
3060 FOR I=2 TO N6
3070 LET P(I)=P(1)+D6*(I-1)
3080 LET Q(I)=L6*(I-1)/(N6-1)
3090 NEXT I
3095 LET M5=M5+(M4-M5)*(M5=0)
3100 LET A6=FNA(1.1*M5)+1
3105 LET A6=A6*(A6>0)+(A6<=0)+(M4 0)
3110 LET B6=FNA(1.1*D6)
3115 LET B6=-1*B6*(B6<0)
3120 LET B6=B6-(B6=0)
3125 LET W3=7*(A6+B6+1)
3130 LET T6=INT(2.9*W3*(N6-2)/L6)+1
3132 REM E6 AND CORRECTIONS TO W3 OVERCOME BUGS IN "NUMBER"
3134 LET E6 = 10^(FNA(1.1*D6))
3136 LET E6 = E6 + (1-E6)*(D6>=1)
3138 LET L8 = 5*(1+(G6=2))
3140 FOR I=1 TO N6
3141 IF P(I)=0 THEN 3144
3142 NEXT I
3143 LET I=1
3144 LET SO=I-T6*INT((I-1)/T6)
3146 ON R6 GO TO 3150,3266
3150 LET X0 = P(1)
3160 LET X9 = P(N6)
3170 PLOT Q(2)/2,L8*A2,3
3172 PLOT Q(2)/2,0,2
3180 FOR I=2 TO N6-1
3182 PLOT Q(I),10*A2,3
3184 PLOT Q(I),0,2
3186 PLOT (Q(I)+Q(I+1))/2,L8*A2,3
3188 PLOT (Q(I)+Q(I+1))/2,0,2
3190 NEXT I
3196 FOR I=SO TO N6-1 STEP T6
3197 IF P(I)=0 THEN 3206
3198 LET W3 = FNW(P(I),D6)
3199 LET W3 = W3 + 10*(P(I)<1)*(P(I)>0) + 7*(P(I)=10)
3200 NUMBER Q(I)-W3,-30,2,P(I)-E6*(P(I)<0),0,B6
3203 GO TO 3210
3206 NUMBER Q(I)-7,-30,2,0,0,-1
3210 NEXT I
3216 PLOT Q(2)/2,M6,3
3218 PLOT Q(2)/2,M6-L8*A2,2
3220 FOR I=2 TO N6-1
3230 PLOT Q(I),M6,3
3240 PLOT Q(I),M6-10*A2,2
3241 PLOT (Q(I)+Q(I+1))/2,M6,3
3242 PLOT (Q(I)+Q(I+1))/2,M6-L8*A2,2
3250 NEXT I
3260 GO TO 3380
3266 PLOT L8*A2,Q(2)/2,3
3268 PLOT Q,Q(2)/2,2
```

APPENDIX I

PROGRAM LISTING - PLOTTR

```
3270 FOR I=2 TO N6-1
3272 PLOT 10*A2,Q(I),3
3274 PLOT 0,Q(I),2
3276 PLOT L8*A2,(Q(I)+Q(I+1))/2,3
3278 PLOT 0,(Q(I)+Q(I+1))/2,2
3280 NEXT I
3282 IF W3 < 25 THEN 3300
3284 FOR I=S0+T6*((S0=1)-P(1)=0)) TO N6-1 STEP T6
3286 IF P(I)=0 THEN 3296
3289 LET W3 = FNW(P(I),D6)
3291 LET W3 = W3 + 10*(P(I)<1)*(P(I)>0) + 7*(P(I)=10)
3292 NUMBER -16,Q(I)-W3,2,P(I)-E6*(P(I)<0),1,B6
3294 GO TO 3298
3296 NUMBER -16,Q(I)-7,2,0,1,-1
3298 NEXT I
3299 GO TO 3316
3300 FOR I=2 TO N6-1
3302 IF P(I)=0 THEN 3312
3304 LET W3 = FNW(P(I),D6)
3307 LET W3 = W3 + 10*(P(I)<1)*(P(I)>0) + 7*(P(I)=10)
3308 NUMBER -27-W3,Q(I)-7,2,P(I)-E6*(P(I)<0),0,B6
3310 GO TO 3314
3312 NUMBER -27,Q(I)-7,2,0,0,-1
3314 NEXT I
3316 PLOT M6,Q(2)/2,3
3318 PLOT M6-L8*A2,Q(2)/2,2
3320 FOR I=2 TO N6-1
3330 PLOT M6,Q(I),3
3340 PLOT M6-10*A2,Q(I),2
3343 PLOT M6,(Q(I)+Q(I+1))/2,3
3346 PLOT M6-L8*A2,(Q(I)+Q(I+1))/2,2
3350 NEXT I
3360 LET Y0 = P(1)
3370 LET Y9 = P(N6)
3380 RETURN
3490 REM THIS SUBROUTINE SETS AXES SCALES FOR LOGARITHMIC PLOTS
3500 FOR I=1 TO 100
3510 LET P(I) = 1.0E6
3520 NEXT I
3530 LET G6 = 10 - INT(10^(M4-INT(M4)))
3540 LET G6 = G6 - 9*(G6=9)
3550 LET H6 = INT(10 (M5-INT(M5)))
3560 LET H6 = H6 - 9*(H6=9)
3570 LET D6 = INT(M5) - INT(M4) + (G6=0) + (H6=0)
3580 LET N6 = 9*D6 + G6 + H6 - 8
3590 FOR I=G6+1 TO N6 STEP 9
3600 LET P(I) = INT(M4) + (G6>0) + (I-G6-1)/9
3610 NEXT I
3620 LET Z6 = L6/(D6-1+FNL((H6+1)/(1-G6/10)))
3630 FOR I=1 TO G6+1
3640 LET Q(I) = Z6*(FNL((9-G6+I)/(10-G6)))
3650 NEXT I
3660 FOR J=0 TO D6-1
3670 FOR I=G6+2 TO G6+10
3680 LET Q(I+9*J) = Q(G6+1) + Z6*(J+FNL(I-G6))
3690 NEXT I
3700 NEXT J
3710 IF D6>=2 THEN 3780
3720 FOR I=1 TO G6
3730 LET P(I) = 109 - G6 + I
3733 LET P(I) = P(I) + 1E6*(P(I)=107)
3737 LET P(I) = P(I) + 1E6*(P(I)=109)
3740 NEXT I
3750 FOR I=G6+2 TO N6
3760 LET P(I) = I - G6 + 100
```

APPENDIX I

PROGRAM LISTING - PLOTTR

```
3763 LET P(I) = P(I) + 1E6*(P(I)=107)
3767 LET P(I) = P(I) + 1E6*(P(I)=109)
3770 NEXT I
3780 ON R6 GO TO 3790,3950
3790 FOR I=2 TO N6-1
3800 PLOT Q(I),10*(1+(P(I)<100)),3
3810 PLOT Q(I),0,2
3812 NEXT I
3815 FOR I=1 TO N6
3820 IF (P(I)>1000) THEN 3880
3820 IF (P(I)>100) THEN 3870
3840 NUMBER Q(I)-20,-40,2,10,0,-1
3850 NUMBER Q(I),-21,2,P(I)-(P(I)<0),0,-1
3860 GO TO 3880
3870 NUMBER Q(I)-5,-28,2,P(I)-100,0,-1
3880 NEXT I
3890 FOR I=2 TO N6-1
3900 PLOT Q(I),M6-10*(1+(P(I)<100)),3
3910 PLOT Q(I),M6,2
3920 NEXT I
3930 LET X0 = P(G6+1) + FNL (10-G6) - 1
3940 LET X9 = P(9*D6+G6-8) + FNL (H6+1)
3945 GO TO 4110
3950 FOR I=1 TO N6
3960 PLOT 10*(1+(P(I)<100)),Q(I),3
3970 PLOT 0,Q(I),2
3980 IF (P(I)>1000) THEN 4040
3990 IF (P(I)>100) THEN 4030
4000 NUMBER -46,Q(I)-16,2,10,0,-1
4010 NUMBER -26,Q(I)+2,2,P(I)-(P(I)<0),0,-1
4020 GO TO 4040
4030 NUMBER -36,Q(I)-7,2,P(I)-100,0,-1
4040 NEXT I
4050 FOR I=2 TO N6-1
4060 PLOT M6-10*(1+(P(I)<100)),Q(I),3
4070 PLOT M6,Q(I),2
4080 NEXT I
4090 LET Y0 = P(G6+1) + FNL(10-G6)-1
4100 LET Y9 = P(9*D6+G6-8) + FNL(H6+1)
4110 RETURN
4950 PRINT"DO YOU WISH TO PRINT EXTRA LINES OR LABELS? 0=NO;1=YES";
4960 INPUT Z8
4970 IF Z8=0 THEN 5070
4980 PRINT"HOW MANY LINES (60 CHARACTERS MAXIMUM) ARE DESIRED";
4990 INPUT Z9
5000 PRINT"DO YOU WISH TO HAVE SYMBOLS PRINTED OUT? 0=NO;1=YES";
5002 INPUT Z8
5004 IF Z8=0 THEN 5020
5006 FOR M=0 TO 2
5008 FOR N=0 TO 4
5010 PLOT 50*N+250*M,800,1
5012 MARKER N+1
5014 NEXT N
5016 NEXT M
5020 PRINT"THE DESIRED ALPHANUMERIC DATA WILL NOW BE ENTERED BETWEEN"
5021 PRINT"QUOTATION MARKS"
5034 FOR I=1 TO Z9
5035 PRINT"THE DATA FOR LINE #";I;"IS";
5036 INPUT C$
5037 LET C1 = LEN(C$)
5038 SYMBOL 0,765-35*I,2,C$,0,C1
5050 PRINT
5060 NEXT I
5064 RSTR
5070 STOP
5080 END
```

APPENDIX II

FLOW CHART - MAIN PROGRAM



APPENDIX II

FLOW CHART - MAIN PROGRAM



APPENDIX II

FLOW CHART - MAIN PROGRAM



APPENDIX II

FLOW CHART - MAIN PROGRAM



APPENDIX II

FLOW CHART MAIN PROGRAM



APPENDIX II

FLOW CHART - MAIN PROGRAM



APPENDIX I

FLOW CHART - MAIN PROGRAM



APPENDIX II

FLOW CHART - MAIN PROGRAM



APPENDIX II

FLOW CHART - MAIN PROGRAM



APPENDIX II

FLOW CHART - MAIN PROGRAM



APPENDIX II

FLOW CHART - MAIN PROGRAM



APPENDIX II

FLOW CHART - SP1



APPENDIX II

FLOW CHART - SP1



APPENDIX I

FLOW CHART - SPI



APPENDIX II

FLOW CHART - SF1



APPENDIX

FLOW CHART - SF1



APPENDIX II

FLOW CHART - SP1



APPENDIX I

FLOW CHART - SP1



APPENDIX II

FLOW CHART - SP1



FLOW CHART - SP2



FLOW CHART - SP2



APPENDIX I

FLOW CHART - SP2



APPENDIX II

FLOW CHART - SUB-



APPENDIX II

FLOW CHART - SP2



APPENDIX II

FLOW CHART OF P1



APPENDIX II

FLOW CHART



APPENDIX III

SIMPLIFIED OPERATING INSTRUCTIONS

This set of instructions assumes that the MS operator has loaded PLOTTR into the BASIC complier and that the plotter is on and the pen located at the bottom of the paper over a page divider.

1. The program execution is initiated* by entering the command RUN.

2. The label for the X-axis is entered between quotes. e.g. "X-axis".

3. The label for the Y-axis is entered in a similar fashion.

4. The number of sets of data (up to a maximum of five) is entered.

5. Selections are made for each data set for:

   (a) the symbol to be used for plotting a point based on: 0 = no symbol; 1 = crosses; 2 = X's; 3 = diamonds; 4 = squares and 5 = double diamonds.

   (b) the type of line to be drawn on the graph based on: 0 = no line required; 1 = individual points joined by straight line segments; 2 = a straight line resulting from a least squares treatment of the points.

   (c) the appearance of the line based on: 0 = solid (option 1 in (b) automatically assumes a solid line); 1 = short dashed; 3 = long dashed. The selection is made by typing in the choice for (a), followed by a comma; the choice for (b) followed by a comma and the choice for (c).

6. The size of the graph is selected based on: 1 = half-page blue-line; 2 = full-page blue-line.

7. Selections are made for the format of each axis on the basis of: 1 = linear; 2 = logarithmic (base 10) or 3 = inverse, in the form of: choice for x-axis, comma, choice for y-axis.

8. The data (up to 100 points) are entered for each set, in turn, in the form: x, comma, y. To delete an error made earlier in this set, (999999, 0) is entered the appropriate number of times to 'backspace' to the incorrect value, and then data entry is resumed.

   When entry of the data for a set is completed, the end is signalled by entering the pair, (999999, 0). The next data set may now be entered.

9. After the graph has been completed, the option of plotting additional copies is provided based on: 0 = no copy; 1 - additional copy desired.

* All data is entered into the computer by typing in the required numbers or letters and ending the operation by pressing RETURN.

10. The option of printing extra lines, labels or plotting symbols is provided based on: 0 = no further output; 1 = output desired. In the event output is desired, the number of lines of print is then entered followed by the choice of printing out a line of plotting symbols for inclusion in captions. The lines of alphanumeric characters (60 per line) are then entered between quotes, e.g. "XYZ", and printed out by the plotter. The process is repeated for each subsequent line of material.

11. The program ends.

UNCLASSIFIED

Security Classification

DOCUMENT CONTROL DATA – R & D

(Security classification of title, body of abstract and indexing annotation must be entered when the overall document is classified)

1. ORIGINATING ACTIVITY

Defence Research Establishment Ottawa

2a. DOCUMENT SECURITY CLASSIFICATION

UNCLASSIFIED

2b. GROUP

3. DOCUMENT TITLE

A Computer Program in Finnigan BASIC for Plotting Report-Quality Graphs Using a Finnigan 6115 Data System

4. DESCRIPTIVE NOTES (Type of report and inclusive dates)

5. AUTHOR(S) (Last name, first name, middle initial)

Purdon, J. Garfield and Preston, Jonathan M.

6. DOCUMENT DATE	7a. TOTAL NO. OF PAGES	7b. NO. OF REFS
	42	2

8a. PROJECT OR GRANT NO.	9a. ORIGINATOR'S DOCUMENT NUMBER(S)
	DREO Technical Note 82-3
8b. CONTRACT NO.	9b. OTHER DOCUMENT NO.(S) (Any other numbers that may be assigned this document)

10. DISTRIBUTION STATEMENT

Unlimited distribution

11. SUPPLEMENTARY NOTES	12. SPONSORING ACTIVITY

13. ABSTRACT

A BASIC computer program has been developed for drawing either half- or full-page-size report-quality graphs of externally generated data on a Finnigan 4010 Gas Chromatograph/Mass Spectrometer Data System. Up to five sets of results can be linearly, logarithmically or inversely presented. As well as drawing the borders and printing labels supplied by the user, the program selects appropriate divisions for the axes and scales the data accordingly. Either solid straight lines joining all points or solid or dashed straight lines resulting from an internal linear least squares calculation can be included on the graph. Provision has also been made for the inclusion of threshold lines on the graph if desired and for the separate printing of lines of alphanumeric and plotting characters for use in captions or headings. A listing, flow-chart and set of simplified instructions for its use are included.

DSIS

72-508

KEY WORDS

BASIC computer program

INSTRUCTIONS

1. ORIGINATING ACTIVITY: Enter the name and address of the organization issuing the document.

2a. DOCUMENT SECURITY CLASSIFICATION: Enter the overall security classification of the document including special warning terms whenever applicable.

2b. GROUP: Enter security reclassification group number. The three groups are defined in Appendix 'M' of the DRB Security Regulations.

3. DOCUMENT TITLE: Enter the complete document title in all capital letters. Titles in all cases should be unclassified. If a sufficiently descriptive title cannot be selected without classification, show title classification with the usual one-capital-letter abbreviation in parentheses immediately following the title.

4. DESCRIPTIVE NOTES: Enter the category of document, e.g. technical report, technical note or technical letter. If appropriate, enter the type of document, e.g. interim, progress, summary, annual or final. Give the inclusive dates when a specific reporting period is covered.

5. AUTHOR(S): Enter the name(s) of author(s) as shown on or in the document. Enter last name, first name, middle initial. If military, show rank. The name of the principal author is an absolute minimum requirement.

6. DOCUMENT DATE: Enter the date (month, year) of Establishment approval for publication of the document.

7a. TOTAL NUMBER OF PAGES: The total page count should follow normal pagination procedures, i.e., enter the number of pages containing information.

7b. NUMBER OF REFERENCES: Enter the total number of references cited in the document.

8a. PROJECT OR GRANT NUMBER: If appropriate, enter the applicable research and development project or grant number under which the document was written.

8b. CONTRACT NUMBER: If appropriate, enter the applicable number under which the document was written.

9a. ORIGINATOR'S DOCUMENT NUMBER(S): Enter the official document number by which the document will be identified and controlled by the originating activity. This number must be unique to this document.

9b. OTHER DOCUMENT NUMBER(S): If the document has been assigned any other document numbers (either by the originator or by the sponsor), also enter this number(s).

10. DISTRIBUTION STATEMENT: Enter any limitations on further dissemination of the document, other than those imposed by security classification, using standard statements such as:

    (1) "Qualified requesters may obtain copies of this document from their defence documentation center."

    (2) "Announcement and dissemination of this document is not authorized without prior approval from originating activity."

11. SUPPLEMENTARY NOTES: Use for additional explanatory notes.

12. SPONSORING ACTIVITY: Enter the name of the departmental project office or laboratory sponsoring the research and development. Include address.

13. ABSTRACT: Enter an abstract giving a brief and factual summary of the document, even though it may also appear elsewhere in the body of the document itself. It is highly desirable that the abstract of classified documents be unclassified. Each paragraph of the abstract shall end with an indication of the security classification of the information in the paragraph (unless the document itself is unclassified) represented as (TS), (S), (C), (R), or (U).

    The length of the abstract should be limited to 20 single-spaced standard typewritten lines; 7½ inches long.

14. KEY WORDS: Key words are technically meaningful terms or short phrases that characterize a document and could be helpful in cataloging the document. Key words should be selected so that no security classification is required. Identifiers, such as equipment model designation, trade name, military project code name, geographic location, may be used as key words but will be followed by an indication of technical context.